

Using the I/Os in the HT47R20A-1

D/N : HA0033E

Introduction

The HT47R20A-1 has an 8-bit dual input/output port PA and a 4-bit input port PB, corresponding to RAM address [12H] and [14H] respectively. The highest 4-bits of PA are NMOS input/outputs while the lower 4-bits of PA can be set as NMOS input/output or CMOS outputs using configuration options. The lower 4-bits of PA can also be configured to have pull-high resistors. Every PA pin has a wake-up function enabled by configuration options. Port PB can only be used as inputs and has pull-high resistors. For PB and when PA is used as inputs, note that the input data does not have a latch function. When PA is used as outputs, all data will be latched until new values are written into the output latch.

When any pin on Port PA is to be used as an input, it is important to know that a $\bar{1}$ must be written to the corresponding bit. This will set the output high which in effect turns off the corresponding NMOS output driver. This can be achieved by executing a "SET [m].i" instruction to turn off the NMOS driver, after which a "MOV A,[m]" instruction can be executed to read the data in the register. These two instructions must be together in case that pin values will be changed because of the "Read-Modify-Write" process.


Some output instructions will first read data from the whole port before executing the write to the output latch. This is known as a "Read-Modify-Write" process. For example, the "SET [m].i", "CLR [m].i", "CPL [m]" and "CPLA [m]" will read the whole port status into the CPU, then execute the required operations before writing the final result back out to the latch or the accumulator. It is important to note that this process may change the original value of the internal latch.

I/O Port Usage

This application uses a 4-LED scanning function to demonstrate how to use the I/Os on the HT47R20A-1 device.

- Hardware

In this application pins PA4~PA7 are connected to LEDs as shown below.



Note that the highest 4-bits on PA are NMOS input/outputs with pull-high resistors. When used as outputs, their drive capability is when the outputs are low. Therefore when connecting to the LEDs, the LED cathode should be connected to the PA high bit port and the anode to V_{DD} via a current limiting resistor.

- Software

The following program uses the timer/counter interrupt to control the amount of time the LEDs are illuminated. The clock source is the system clock with an initial counter value of 00H.

- Program list

```
include ht47r20a-1.inc
data .section 'data'
    count db ?
code .section at 0 'code'
    org 00h
    jmp start
    org 04h
    reti
    org 08h
    reti
    org 0ch
    reti
    org 10h
    jmp timer_int ;timer/counter interrupt routine
                    ;address
```

```

;-----
start:
  clr  intc0
  clr  intc1
  clr  adcr.1      ;timer/counter enabled
  mov  a,08h      ;set the system clock as the clock source
                        ;for the timer/counter

  mov  tmrc,a
  mov  a,0eeh
  mov  count,a    ;use count to control the LEDs
  mov  a,00h      ;set the timer/counter initial value
  mov  tmrbl,a
  mov  tmrbh,a
  mov  tmral,a
  mov  tmrah,a
  set  tmrc.4     ;start the timer/counter
  set  intc1.0    ;timer/counter interrupt enabled
  set  intc0.0    ;global interrupt enabled
  jmp  $

;-----
timer_int:          ;timer/counter interrupt service
                  ;subroutine

  rr   count
  mov  a,count     ;illuminate the next LED
  mov  pa,a
  reti

```

A Description of the I/O Port "Read-Modify-Write"

The following program demonstrates the "Read-Modify-Write" process. Note that here PA.0 and PA.7 are connected together.

- Program list

```

include ht47r20a-1.inc
code .section at 0 'code'
  org  00h
  jmp  start
  org  04h
  reti
  org  08h
  reti
  org  0ch
  reti
  org  10h
  reti

;-----
start:          ;PA.7(latch value)pin value,
                ;PA.0(latch value)pin value

```

```
      clr   pa           ; (0)0 (0)0
      set   pa.7 ;(1)    ; (1)0 (0)0
      set   pa.0 ;(2)    ; (0)0 (1)0
      set   pa.7 ;(3)
      sz    pa.7
      jmp   read_1
read_0:
      jmp   $
read_1:
      jmp   $
```

- Program Description

Remembering that the two pins are connected together, when executing the above program it is discovered that the level of pins PA.0 and PA.7 remains low. However the value of the PA.0 and PA.7 internal latch has changed. Please refer to the program notes.

By using the single-step function the true action of the "Read-Modify-Write" process can be seen. When step (2) of the program has ended, and if the connecting wire between pins PA.0 and PA.7 is removed, then pin PA.0 will go to a high level after the execution of the next statement.

When statement (3) is added, after the execution of statement (2), if the connecting wire is removed the final result is PA.0=1 and PA.7=1. If the connecting wire is removed after statement (3) is executed, the final result is PA.0=0 and PA.7=1.

It is important that the differences in the examples above are carefully noted.